# $Id: PKGBUILD 70022 2012-04-28 09:24:14Z allan $
# Maintainer: Eric Bélanger <eric@archlinux.org>

pkgname=hardinfo
pkgver=0.5.1
pkgrel=5
pkgdesc="A system information and benchmark tool."
arch=('i686' 'x86_64')
url="http://hardinfo.berlios.de/wiki/index.php/Main_Page"
license=('GPL2')
depends=('gtk2' 'libsoup')
options=('!emptydirs')
source=(http://download.berlios.de/hardinfo/${pkgname}-${pkgver}.tar.bz2 \
        fixsensors.patch hardinfo.distro)
sha1sums=('983f0445aa60e02156bc5a5eaedeffb30b8e4d64'
          'd2151dce76dca238102937d0e15541f172bca530'
          '4c31cbbfbdeb48593641ff600c8d4fb7bda2b01c')

build() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  patch -p1 < ../fixsensors.patch
  sed -i 's|/usr/lib64|/usr/lib|' configure
  ./configure --prefix=/usr
  sed -i 's|lib64|lib|' binreloc.c
  sed -i 's|{ DB_PREFIX "debian_version",	"deb"  },|{ DB_PREFIX "hardinfo.distro",       "arch"  },{ DB_PREFIX "debian_version",	"deb"  },|' computer.h
  make
}

package() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  make DESTDIR="${pkgdir}" install
  install -D -m644 ../hardinfo.distro "${pkgdir}/etc/hardinfo.distro"
}
